```
Vec[x_List] := List /@ x;
Vec[x__] := Vec[{x}];

V3 =
  Flatten[Table[If[i == j || j == k || k == i, 0, 1], {i, 1, 3}, {j, 1, 3}, {k, 1, 3}], 1] /
   Sqrt[2];
AscendingOp[x_] := Transpose[V3].KroneckerProduct[x, IdentityMatrix[3]].V3;
FusionOp[x_, y_] := Transpose[V3].KroneckerProduct[x, y].V3;
MatrixBasis[i_, j_] := Array[Boole[#1 == i && #2 == j] &, {3, 3}];
HS[x_, y_] := Tr[ConjugateTranspose[x].y]

Transpose[V3].V3 // MatrixForm
```

$$\begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix}$$

```
asconbas = AscendingOp[MatrixBasis[First@#, Last@#]] & /@ Tuples[{1, 2, 3}, 2];

bas = MatrixBasis[First@#, Last@#] & /@ Tuples[{1, 2, 3}, 2];

ascmat = Table[HS[bas[[i]], asconbas[[j]]], {i, 1, 9}, {j, 1, 9}];

{eigenval, eigenveccoeff} = Eigensystem[ascmat];

eigenvecs = #.bas & /@ eigenveccoeff;

eigenval
```

$$\left\{1, -\frac{1}{2}, -\frac{1}{2}, -\frac{1}{2}, -\frac{1}{2}, -\frac{1}{2}, \frac{1}{2}, \frac{1}{2}, \frac{1}{2}\right\}$$

```
MatrixForm /@ eigenvecs
```

$$\left\{\begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix}, \begin{pmatrix} -1 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 1 \end{pmatrix}, \begin{pmatrix} 0 & 0 & 0 \\ 0 & 0 & -1 \\ 0 & 1 & 0 \end{pmatrix}, \begin{pmatrix} 0 & 0 & -1 \\ 0 & 0 & 0 \\ 1 & 0 & 0 \end{pmatrix},\right.$$
$$\left.\begin{pmatrix} -1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 0 \end{pmatrix}, \begin{pmatrix} 0 & -1 & 0 \\ 1 & 0 & 0 \\ 0 & 0 & 0 \end{pmatrix}, \begin{pmatrix} 0 & 0 & 0 \\ 0 & 0 & 1 \\ 0 & 1 & 0 \end{pmatrix}, \begin{pmatrix} 0 & 0 & 1 \\ 0 & 0 & 0 \\ 1 & 0 & 0 \end{pmatrix}, \begin{pmatrix} 0 & 1 & 0 \\ 1 & 0 & 0 \\ 0 & 0 & 0 \end{pmatrix}\right\}$$

```
eigenvecsbas = Table[HS[bas[[i]], eigenvecs[[j]]], {i, 1, 9}, {j, 1, 9}];

fusonbas = Table[FusionOp[eigenvecs[[i]], eigenvecs[[j]]], {i, 1, 9}, {j, 1, 9}];

fusonbasbas = Table[HS[bas[[k]], fusonbas[[i, j]]], {i, 1, 9}, {j, 1, 9}, {k, 1, 9}];

fusioncoefficients =
  Table[LinearSolve[eigenvecsbas, fusonbasbas[[i, j]]], {i, 1, 9}, {j, 1, 9}];
```

```
fusioncoefficients[[8]] // MatrixForm
```

$$\begin{pmatrix} 0 & 0 & 0 & 0 & 0 & 0 & 0 & \frac{1}{2} & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & \frac{1}{2} & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & \frac{1}{2} & 0 \\ 0 & 0 & \frac{1}{2} & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & \frac{1}{2} \\ \frac{1}{3} & -\frac{1}{3} & 0 & 0 & \frac{2}{3} & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & \frac{1}{2} & 0 & 0 \end{pmatrix}$$

```
Nmat = Map[Boole@*UnequalTo[0], fusioncoefficients, {3}];
```

```
SymmetricMatrixQ@Nmat[[#]] & /@ Range@9
```

```
{True, False, True, False, False, False, True, False, False}
```

```
FusionOp[eigenvecs[[2]], eigenvecs[[6]]] // MatrixForm
```

$$\begin{pmatrix} 0 & \frac{1}{2} & 0 \\ -\frac{1}{2} & 0 & 0 \\ 0 & 0 & 0 \end{pmatrix}$$

```
checkComm[n_, idx_] :=
  And @@ Flatten[Table[n[[i, j, k]] == n[[j, i, k]], {i, idx}, {j, idx}, {k, idx}]];
checkAssoc[n_, idx_] := And @@ Flatten[Table[Sum[n[[i, j, k]] n[[k, l, m]], {k, idx}] ==
      Sum[n[[j, l, k]] n[[i, k, m]], {k, idx}], {i, idx}, {j, idx}, {l, idx}, {m, idx}]];
```

```
checkComm[Nmat, {1, 2, 3, 4, 5, 6, 7, 8, 9}]
```

```
True
```

```
checkAssoc[Nmat, {1, 2, 5}]
```

```
False
```