The Fibonacci Chain

First load the TriCats package and load the standard library to access oft-used diagrams:

```
SetDirectory[NotebookDirectory[]];
If[FileExistsQ@#, SetDirectory[(Get@#)@$UserName]] &@"packagelocation";
<< TriCats`;
LoadLibrary["stdlib"];
```

Load diagrams:
- C4Atoms denotes the list of zero- and one-faced diagrams in C_4;
- cap and cup denote the obvious diagrams;
- line denotes a straight line diagram;
- trivalentb (short for “trivalent branch”) denotes the trivalent vertex with one ingoing and two outgoing legs.

```
Move2Down = DiagramMoveDown[#, 2] &;
C4Atoms = Move2Down /@ Retrieve["stdlib:C4Atoms"];
cap = Retrieve["stdlib:Cap"];
cup = Retrieve["stdlib:Cup"];
line = Retrieve["stdlib:Line"];
trivalentb = Retrieve["stdlib:TrivalentBranch"];
```

```
so3qbasisrule = x_Diagram /; IsomorphicDiagramQ[x, C4Atoms[[4]]] →
   {-1/(d - 1), 1/(d - 1), 1}.C4Atoms[[1 ;; 3]];
fibdt = {d → 1/2 (1 + √5), t → 1/2 (1 - √5), b → 1};
fibbasisrule =
  x_Diagram /; IsomorphicDiagramQ[x, C4Atoms[[3]]] → {1, -1/d}.C4Atoms[[1 ;; 2]];
fibbeta = FullSimplify[Exp[ⅈ (n - 1)/n π] /. n → 5];
fibbraiding = {1, fibbeta}.C4Atoms[[1 ;; 2]];
```

```
YangBaxter[braiding_, line_] :=
 {DiagramCompose[DiagramTensor[braiding, line], DiagramTensor[line, braiding],
   DiagramTensor[braiding, line]], DiagramCompose[DiagramTensor[line, braiding],
   DiagramTensor[braiding, line], DiagramTensor[line, braiding]]}
```

```
VTensor[braiding_, trivalentb_, line_, cup_, cap_] :=
  DiagramCompose[DiagramTensor[line, trivalentb, cup],
   DiagramTensor[braiding, DiagramConjugate[braiding], line],
   DiagramTensor[line, trivalentb, line, line, line],
   DiagramTensor[line, line, cap, line, line]];
```

```
AscendingOp[v_, line_] := Function[x,
   DiagramCompose[v, DiagramTensor[x, line, line], DiagramConjugate[v]]];
```

```
FusionOp[v_, line_] :=
 Function[{x, y}, DiagramCompose[v, DiagramTensor[x, y], DiagramConjugate[v]]]
```

Braiding test

First we check that fibbraiding is indeed a braiding.
Biunitarity:

```
FullSimplify[
  ReduceDiagram[DiagramCompose[DiagramConjugate[fibbraiding], fibbraiding],
    dimC4 → 2] /. fibdt] // EnsureGraph
```

Diagram[

, {1, 2}, {3, 4}]

```
FullSimplify[ReduceDiagram[DiagramCompose[DiagramTensor[line, cup, line],
     DiagramTensor[DiagramConjugate[fibbraiding], fibbraiding],
     DiagramTensor[line, cap, line]], dimC4 → 2] /. fibdt] // EnsureGraph
```

Diagram[

, {1, 2}, {3, 4}]

```
fibcheckYB = ReduceDiagram[Subtract @@ YangBaxter[fibbraiding, line], dimC4 → 2];
```

```
FullSimplify[Components[fibcheckYB, DistinctDiagrams[fibcheckYB]] /. fibdt]
```

{0, 0, 0, 0, 0}

The chain example

```
fibv = ReduceDiagram[VTensor[fibbraiding, trivalentb, line, cup, cap], dimC4 → 2];
```

```
fibascending = AscendingOp[fibv, line];
```

```
fibvmatrix = RootReduce@Transpose[
   FullSimplify[
      Components[ReduceDiagram[fibascending[#], b → 1, dimC4 → 2] //.
         {so3qbasisrule, fibbasisrule} //. fibdt, C4Atoms[[1 ;; 2]]]
     ] & /@ C4Atoms[[1 ;; 2]]
  ]
```

$$\left\{\left\{1, \frac{1}{2}\left(3-\sqrt{5}\right)\right\}, \left\{0, \frac{1}{2}\left(3-\sqrt{5}\right)\right\}\right\}$$

```
{fibveigenvalues, fibveigenvectors} = Eigensystem[fibvmatrix] // RootReduce
```

$$\left\{\left\{1, \frac{1}{2}\left(3-\sqrt{5}\right)\right\}, \left\{\{1, 0\}, \left\{\frac{1}{2}\left(1-\sqrt{5}\right), 1\right\}\right\}\right\}$$

```
-Log2[#] & /@ fibveigenvalues // N
```

{0., 1.38848}

```
fibfusion = FusionOp[fibv, line];
```

```
fibfusionmatrix = FullSimplify[
  Table[Components[ReduceDiagram[fibfusion[fibveigenvectors⟦i⟧.C4Atoms⟦1 ;; 2⟧,
        fibveigenvectors⟦j⟧.C4Atoms⟦1 ;; 2⟧], b → 1, dimC4 → 3] //.
      {so3qbasisrule, fibbasisrule} //. fibdt, C4Atoms⟦1 ;; 2⟧],
   {i, 1, 2}, {j, 1, 2}]]
```

$$\left\{\left\{\{1, 0\}, \left\{2-\sqrt{5}, \frac{1}{2}\left(3-\sqrt{5}\right)\right\}\right\}, \left\{\left\{2-\sqrt{5}, \frac{1}{2}\left(3-\sqrt{5}\right)\right\}, \left\{\frac{1}{2}\left(11-5\sqrt{5}\right), 5-2\sqrt{5}\right\}\right\}\right\}$$

The fusion coefficients $f_{\gamma}^{\alpha\beta}$:

```
fibfusioncoefficients = Table[
  FullSimplify[LinearSolve[Transpose@fibveigenvectors, fibfusionmatrix⟦i, j⟧]],
  {i, 1, 2}, {j, 1, 2}]
```

$$\left\{\left\{\{1, 0\}, \left\{0, \frac{1}{2}\left(3-\sqrt{5}\right)\right\}\right\}, \left\{\left\{0, \frac{1}{2}\left(3-\sqrt{5}\right)\right\}, \left\{-2+\sqrt{5}, 5-2\sqrt{5}\right\}\right\}\right\}$$

The fusion matrix $N_{\gamma}^{\alpha\beta}$:

```
fibN = Map[Boole@FullSimplify[# ≠ 0] &, fibfusioncoefficients, {3}]
```

{{{1, 0}, {0, 1}}, {{0, 1}, {1, 1}}}

```
fibN⟦1, ;;⟧ // MatrixForm
```

$$\begin{pmatrix} 1 & 0 \\ 0 & 1 \end{pmatrix}$$

```
fibN⟦2, ;;⟧ // MatrixForm
```

$$\begin{pmatrix} 0 & 1 \\ 1 & 1 \end{pmatrix}$$